Example Technology Exercise 15 James Jones

Perform some initial setup

```
(%i1) load("vect")$
      Curl(F):=ev(express(curl(F)),diff)$
      Div(F):=ev(express(div(F)),diff)$
      Grad(f):=ev(express(grad(f)),diff)$
      norm(u):=sqrt(u.u)$
```

1 Problem 15.1.57

Define the vector field F and find the CURL

```
(%i6) F:[x*y^2*z^2,x^2*y*z^2,x^2*y^2*z]$
      ev(express(curl(F)),diff);
```

(%o7) $[0,0,0]$

Since the CURL is the zero vector,
the field is conservative and has
a potential function

```
(%i8) integrate(F[1],x);
      integrate(F[2],y);
      integrate(F[3],z);
```

(%o8) $\frac{x^2 y^2 z^2}{2}$

(%o9) $\frac{x^2 y^2 z^2}{2}$

(%o10) $\frac{x^2 y^2 z^2}{2}$

In this case, all three integrals are the same
there is no combining of terms that is necessary

2 Problem 15.4.16

Define M dx and N dy

(%i11) M:2*atan(y/x);
N:log(x^2+y^2);

(%o11) $2\,\mathrm{atan}\left(\frac{y}{x}\right)$

(%o12) $\log\left(y^2+x^2\right)$

It's pretty obvious this is not conservative, right?
But check anyway because if it is, then the problem
is really easy.

(%i13) diff(M,y)-diff(N,x),factor;

(%o13) 0

Wow, what do you know? It is conservative!
That means the line integral is 0

3 Surface Integral

Find the surface integral for sqrt(x^2+y^2+z^2) dS
Surface is portion of 4x+3y+2z=24 in first octant

Define the function and the integrand

(%i14) f:4*x+3*y+2*z-24;
integrand:x^2+y^2+z^2;

(%o14) $2\,z+3\,y+4\,x-24$

(%o15) $z^2+y^2+x^2$

Declare a helper function to find the dS

(%i16) dS(f,t):=norm(Grad(f/coeff(f,t)))$

3.1 xy-plane, z = 0

Find upper function of region

```
(%i17) boundary:solve(subst(z=0,f),y)[1];
       u:rhs(%)$
       v:rhs(solve(u,x)[1])$
```

(%o17) $y=-\frac{4\,x-24}{3}$

Solve for z and substitute into integrand

```
(%i20) solve(f,z);
       subst(%[1],integrand);
```

(%o20) $[z=-\frac{3\,y+4\,x-24}{2}]$

(%o21) $\frac{{\left(3\,y+4\,x-24\right)}^{2}}{4}+y^{2}+x^{2}$

Integrate

```
(%i22) integrate(integrate(%*dS(f,z),y,0,u),x,0,v);
```

(%o22) $488\,\sqrt{29}$

3.2 yz-plane, x = 0

Find upper function of region

```
(%i23) boundary:solve(subst(x=0,f),z)[1];
       u:rhs(%)$
       v:rhs(solve(u,y)[1])$
```

(%o23) $z=-\frac{3\,y-24}{2}$

Solve for x and substitute into integrand

(%i26) solve(f,x);
subst(%[1],integrand);

(%o26) $[x=-\frac{2z+3y-24}{4}]$

(%o27) $\frac{(2z+3y-24)^2}{16}+z^2+y^2$

Integrate

(%i28) integrate(integrate(%*dS(f,x),z,0,u),y,0,v);

(%o28) $488\sqrt{29}$

3.3 xz-plane, y = 0

Find upper function of region

(%i29) boundary:solve(subst(y=0,f),z)[1];
u:rhs(%)$
v:rhs(solve(u,x)[1])$

(%o29) $z=12-2x$

Solve for y and substitute into integrand

(%i32) solve(f,y);
subst(%[1],integrand);

(%o32) $[y=-\frac{2z+4x-24}{3}]$

(%o33) $\frac{(2z+4x-24)^2}{9}+z^2+x^2$

Integrate

(%i34) integrate(integrate(%*dS(f,y),z,0,u),x,0,v);

(%o34) $488\sqrt{29}$

4 Problem 15.7.13

Define F

```
(%i35) F:[x,y^2,-z];
```

(%o35) $[x,y^2,-z]$

The double integral over the surface of F.N dS
is the triple integral over the solid of div F dV

Find the divergence

```
(%i36) divF:ev(express(div(F)),diff);
```

(%o36) $2\,y$

This problem is best done in cylindrical coordinates

```
(%i37) divFcyl:subst(y=r*sin(theta),%);
```

(%o37) $2\,r\,\sin(\theta)$

Now integrate, remember the extra r
in the integrand when switching to cylindrical coordinates

```
(%i38) integrate(integrate(integrate(r*divFcyl,r,0,5),theta,0,2*%pi),z,0,7);
```

(%o38) 0